// File: /contracts/core/staking/category/FixedAPRStaking.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "../Staking.sol";
import "../../../interfaces/ISimpleRewardPool.sol";
/// @title  FixedAPRStaking 
/// @notice the APR is fixed set by admin
contract FixedAPRStaking is Staking{
    event ChangeFixedAPREvent(address account,uint256 from,uint256 to);
    uint256 public stakingType=0;
    uint256 public fixedAPRPercentage=0;
    using EnumerableSet for EnumerableSet.AddressSet;
    using SafeERC20 for IERC20;
    EnumerableSet.AddressSet private stakeUsers;
    mapping(address => StakeUser) public stakeUsersInfo;
    ISimpleRewardPool public rewardPool;
    struct StakeUser {
        bool exist;
        uint256 stakeBlock;
        uint256 lastSettleBlock;
        uint256 stakeAmount;
        uint256 pendingReward;
        uint256 unbindingBlock;
        uint256 unbindingAmount;
    }
    constructor(Structs.StakingParams memory params,uint256 _fixedAPRPercentage) Staking(params) {
        fixedAPRPercentage=_fixedAPRPercentage;
    }
    modifier userStaked(address user) {
        require(stakeUsersInfo[user].exist,"!S");
        _;
    }
    /// @notice setRewardPool,a SimpleRewardPool must be set to FixedAPRStaking 
    /// @param pool, instance address of SimpleRewardPool
    function setRewardPool(address pool) public atLeastProject returns(bool){
        rewardPool=ISimpleRewardPool(pool);
        return true;
    }
    /// @notice stake
    /// @param amount, amount of stakingToken
    function stake(uint256 amount) public isWhitelisted(msg.sender) notStopped whenNotPaused nonReentrant returns(bool){
        address user=msg.sender;
        require(amount>=minStakeAmount,"!P");
        IERC20 token=IERC20(stakingToken);
        uint256 balance = token.balanceOf(user);
        require(balance>=amount,"!B");
        token.safeTransferFrom(user, address(this), amount);
        if (stakeUsersInfo[user].exist){
            require(_claimRewards(user),"CF");
            StakeUser storage stakeUser=stakeUsersInfo[user];
            stakeUser.stakeAmount=stakeUser.stakeAmount+amount;
            stakeUser.stakeBlock=block.number;
        }else{
            stakeUsersInfo[user].exist=true;
            stakeUsersInfo[user].stakeBlock=block.number;
            stakeUsersInfo[user].lastSettleBlock=block.number;
            stakeUsersInfo[user].stakeAmount=amount;
            stakeUsers.add(user);
        }
        totalStakingAmount=totalStakingAmount+amount;
        emit StakeEvent(user,stakingToken,amount);
        return true;
    }
    /// @notice stake all balance of stakingToken
    function stakeAll() public returns(bool){
        IERC20 token=IERC20(stakingToken);
        uint256 balance = token.balanceOf(msg.sender);
        return stake(balance);
    }
    /// @notice normalUnbound ,no fee be deducted,unbound amount can be withdraw after unbindPeriodBlock
    /// @param amount, amount of stakingToken
    function normalUnbound(uint256 amount) public whenNotPaused nonReentrant userStaked(msg.sender) returns(bool){
        address user=msg.sender;
        StakeUser storage stakeUser=stakeUsersInfo[user];
        require(amount > 0&&stakeUser.stakeAmount>=amount,"!B");
        require(_claimRewards(user),"CF");
        totalStakingAmount=totalStakingAmount-amount;
        stakeUser.stakeAmount=stakeUser.stakeAmount-amount;
        stakeUser.unbindingBlock=block.number;
        stakeUser.unbindingAmount=stakeUser.unbindingAmount+amount;
        emit NormalUnboundEvent(user,stakingToken,amount);
        return true;
    }
    /// @notice normalUnboundAll ,no fee be deducted
    function normalUnboundAll() public returns(bool){
        address user=msg.sender;
        normalUnbound(stakeUsersInfo[user].stakeAmount);
        return true;
    }
    /// @notice instantUnbound ,instantUnboundFeePercentage be deducted from amount,user can receive rest of amount immediately
    /// @param amount, amount of stakingToken
    function instantUnbound(uint256 amount) public whenNotPaused nonReentrant userStaked(msg.sender) returns(bool){
        require(allowedInstantUnbound,"!IS");
        address user=msg.sender;
        StakeUser storage stakeUser=stakeUsersInfo[user];
        require(amount > 0&&stakeUser.stakeAmount>=amount,"!B");
        require(_claimRewards(user),"CF");
        uint256 instantFee=amount*instantUnboundFeePercentage/uint256(10000);
        uint256 transferAmount=amount-instantFee;
        totalStakingAmount=totalStakingAmount-amount;
        stakeUser.stakeAmount=stakeUser.stakeAmount-amount;
        IERC20 token=IERC20(stakingToken);
        token.safeTransfer(user, transferAmount);
        if(instantFee>0){
            token.safeTransfer(feeAddress, instantFee);
        }
        if(stakeUser.stakeAmount==0&&stakeUser.unbindingAmount==0&&stakeUser.pendingReward==0){
            delete stakeUsersInfo[user];
            stakeUsers.remove(user);
        }
        emit InstantUnboundEvent(user,stakingToken,amount);
        return true;
    }
    /// @notice instantUnboundAll ,instantUnboundFeePercentage be deducted from amount,user can receive rest of amount immediately
    function instantUnboundAll() public returns(bool){
        address user=msg.sender;
        instantUnbound(stakeUsersInfo[user].stakeAmount);
        return true;
    }
    /// @notice withdrawUnboundAmount , users can withdraw their unbindingAmount after unbindPeriodBlock
    function withdrawUnboundAmount() public whenNotPaused nonReentrant userStaked(msg.sender) returns(bool){
        address user=msg.sender;
        StakeUser storage stakeUser=stakeUsersInfo[user];
        uint256 amount=stakeUser.unbindingAmount;
        require(amount > 0, 'A0');
        require((block.number-stakeUser.unbindingBlock)>=unboundPeriodBlock,"WF");
        require(_claimRewards(user),"CF");
        stakeUser.unbindingAmount=0;
        IERC20 token=IERC20(stakingToken);
        token.safeTransfer(user,amount);
        if(stakeUser.stakeAmount==0&&stakeUser.pendingReward==0){
            delete stakeUsersInfo[user];
            stakeUsers.remove(user);
        }
        emit WithdrawEvent(user,stakingToken,amount);
        return true;
    }
    function _claimRewards(address user) internal returns(bool){
        StakeUser storage stakeUser=stakeUsersInfo[user];
        uint256 reward=claimableRewards(user);
        if (block.number-stakeUser.lastSettleBlock<payoutIntervalBlock){
            stakeUser.pendingReward=reward;
        }else{
            require(rewardPool.claimRewards(user,reward, feeAddress,commissionPercentage),"CF");
            stakeUser.pendingReward=0;
        }
        stakeUser.lastSettleBlock=block.number;
        totalClaimedReward=totalClaimedReward+reward;
        return true;
    }
    /// @notice claimRewards , user can claim per payoutIntervalBlock
    function claimRewards() public whenNotPaused nonReentrant userStaked(msg.sender) returns(bool){
        address user=msg.sender;
        require(block.number-stakeUsersInfo[user].lastSettleBlock>=payoutIntervalBlock,"!PIB");
        return _claimRewards(user);
    }
    /// @notice claimableRewards , user can claim per payoutIntervalBlock
    /// @param user, stake user address
    /// @return Returns amount of reward
    function claimableRewards(address user) public view returns(uint256){
        StakeUser memory stakeUser=stakeUsersInfo[user];
        if (!stakeUser.exist){
            return 0;
        }
        uint256 thisReward=(block.number-stakeUser.lastSettleBlock)*stakeUser.stakeAmount*fixedAPRPercentage/(yearBlock*uint256(10000));
        return thisReward+stakeUser.pendingReward;
    }
    function _settleAllStakeUser() internal returns (bool){
        for(uint256 i=0;i<stakeUsers.length();i++){
            address userAddress=stakeUsers.at(i);
            if (!stakeUsersInfo[userAddress].exist){
                continue;
            }
            uint256 thisReward=(block.number-stakeUsersInfo[userAddress].lastSettleBlock)*stakeUsersInfo[userAddress].stakeAmount*fixedAPRPercentage/(yearBlock*uint256(10000));
            stakeUsersInfo[userAddress].lastSettleBlock=block.number;
            stakeUsersInfo[userAddress].pendingReward=stakeUsersInfo[userAddress].pendingReward+thisReward;
        }
        return true;
    }
    function getAllUnpaidReward() internal view returns(uint256 totalReward){
        for(uint256 i=0;i<stakeUsers.length();i++){
            address userAddress=stakeUsers.at(i);
            if (!stakeUsersInfo[userAddress].exist){
                continue;
            }
            uint256 thisReward=(block.number-stakeUsersInfo[userAddress].lastSettleBlock)*stakeUsersInfo[userAddress].stakeAmount*fixedAPRPercentage/(yearBlock*uint256(10000));
            totalReward=totalReward+thisReward+stakeUsersInfo[userAddress].pendingReward;
        }
        return totalReward;
    }
    /// @notice setFixedAPRPercentage , admin can change FixedAPR, Before APR be changed all user's rewards should be settled
    /// @param aprPer,apr can be infinity
    function setFixedAPRPercentage(uint256 aprPer) public nonReentrant atLeastProject{
        require(_settleAllStakeUser(),"SAS");
        uint256 origin=fixedAPRPercentage;
        fixedAPRPercentage=aprPer;
        emit ChangeFixedAPREvent(msg.sender,origin,fixedAPRPercentage);
    }
    /// @notice getRewardDebt
    /// @return totalRewardInPool 
    /// @return totalStakingAmount
    /// @return totalUnpaidReward 
    function getRewardDebt() public view returns(uint256,uint256,uint256){
        uint256 totalRewardInPool_=IERC20(stakingToken).balanceOf(address(rewardPool));
        uint256 totalUnpaidReward_=getAllUnpaidReward();
        return (totalRewardInPool_,totalStakingAmount,totalUnpaidReward_);
    }
    /// @notice getStakeUserCount
    function getStakeUserCount() public view returns (uint256){
        return stakeUsers.length();
    }

}
// File: /contracts/libraries/Structs.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;
library Structs{
    struct StakingParams{
        address stakingToken;
        uint256 minStakeAmount;
        uint256 unboundPeriodBlock;
        uint256 instantUnboundFeePercentage;
        address accessControl;
        address feeAddress;
        uint256 commissionPercentage;
        uint256 payoutIntervalBlock;
        bool    allowedInstantUnbound;
    }

}

// File: /contracts/interfaces/ISimpleRewardPool.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

interface ISimpleRewardPool{
    function claimRewards(address,uint256,address,uint256) external returns(bool);
}
// File: /contracts/interfaces/IAdminAccess.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

interface IAdminAccess {
    function hasAdminRole(address) external view returns (bool);
    function addToAdminRole(address) external;
    function removeFromAdminRole(address) external;
    function getOwner() external view returns (address);
}
// File: /contracts/core/staking/Staking.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "../../interfaces/IAdminAccess.sol";
import "@openzeppelin/contracts/security/Pausable.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/access/AccessControlEnumerable.sol";
import "../../libraries/Structs.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";

abstract contract Staking is AccessControlEnumerable,ReentrancyGuard{
    using SafeERC20 for IERC20;
    event PausedEvent(address account);
    event UnPausedEvent(address account);
    event StopEvent(address account);
    event SetPayoutIntervalBlockEvent(address account,uint256 from,uint256 to);
    event SetEachBlockSecondEvent(address account,uint256 from,uint256 to);
    event SetYearBlockEvent(address account,uint256 from,uint256 to);
    event SetFeeAddressEvent(address account,address from,address to);
    event SetCommissionEvent(address account,uint256 from,uint256 to);
    event SetMinStakeAmountEvent(address account,uint256 from,uint256 to);
    event SetUnboundPeriodBlockEvent(address account,uint256 from,uint256 to);
    event SetInstantUnboundFeeEvent(address account,uint256 from,uint256 to);
    event SetAllowInstantUnboundEvent(address account,bool allow);
    event StakeEvent(address staker, address token, uint256 amount);
    event NormalUnboundEvent(address staker, address token, uint256 amount);
    event InstantUnboundEvent(address staker, address token, uint256 amount);
    event WithdrawEvent(address staker, address token, uint256 amount);
    event ClaimRewardsEvent(address staker, address token, uint256 amount);
    event AddUserWhitelistEvent(address operator,address user);
    event RemoveUserWhitelistEvent(address operator,address user);
    event ChangeUserWhitelistSwitchEvent(address operator,bool from,bool to);
    event AddProjectRoleEvent(address operator,address account);
    event removeProjectRoleEvent(address operator,address account);

    /// @notice pause all functions
    bool public paused=false;
    /// @dev ethereum [12,2628000] ,bsc [3,10512000]
    /// @notice set different seconds for different blockchain
    uint256 public eachBlockSecond = 12;
    /// @notice block number for whole year
    uint256 public yearBlock=2628000;
    /// @notice stake token's contract address
    address public  stakingToken;
    /// @notice current total staked
    uint256 public totalStakingAmount;
    /// @notice accumulate total claimed reward
    uint256 public totalClaimedReward;
    /// @notice user claim interval
    uint256 public payoutIntervalBlock;
    /// @notice min stake amount
    uint256 public minStakeAmount;
    /// @notice allow withdraw after unboundPeriodBlock
    uint256 public unboundPeriodBlock;
    /// @notice instant unbound fee
    uint256 public instantUnboundFeePercentage;
    /// @notice fee address to receive fees
    address public feeAddress;
    /// @notice commission fee deduct from user's reward
    uint256 public commissionPercentage;
    /// @notice sherpa global permission role
    IAdminAccess public access;
    /// @notice if whitelist check opened then require user in whitelist
    bool public userWhitelistSwitchOpened;
    mapping(address => bool) public userWhitelist;

    /// @notice PROJECT_ROLE in charge permission of own instance
    bytes32 public constant PROJECT_ROLE = keccak256("PROJECT_ROLE");

    /// @notice stop stake immediately or stop at plan block number
    bool public stopped;
    uint256 public stopAtBlock;

    /// @notice true for allow user to instant unbound
    bool public allowedInstantUnbound;

    /// @notice get all fields
    function getAllFields() public view returns(
        uint256 eachBlockSecond_,
        uint256 yearBlock_,
        address stakingToken_,
        uint256 totalStakingAmount_,
        uint256 totalClaimedReward_,
        uint256 payoutIntervalBlock_,
        uint256 minStakeAmount_,
        uint256 unboundPeriodBlock_,
        uint256 instantUnboundFeePercentage_,
        address feeAddress_,
        uint256 commissionPercentage_,
        bool stopped_,
        uint256 stopAtBlock_,
        bool allowedInstantUnbound_){
        return (
            eachBlockSecond,
            yearBlock,
            stakingToken,
            totalStakingAmount,
            totalClaimedReward,
            payoutIntervalBlock,
            minStakeAmount,
            unboundPeriodBlock,
            instantUnboundFeePercentage,
            feeAddress,
            commissionPercentage,
            stopped,
            stopAtBlock,
            allowedInstantUnbound
        );
    }

    /// @notice if whitelist check opened then require user in whitelist
    modifier isWhitelisted(address user) {
        if (userWhitelistSwitchOpened){
            require(userWhitelist[user],"!UW");
        }
        _;
    }
    modifier whenNotPaused() {
        require(!paused, "PD");
        _;
    }
    modifier whenPaused() {
        require(paused, "!PD");
        _;
    }
    modifier notStopped() {
        require(!stopped&&block.number<stopAtBlock, "SD");
        _;
    }
    modifier onlyOwner() {
        require(access.getOwner() == msg.sender, "!O");
        _;
    }
    modifier atLeastAdmin() {
        require(access.hasAdminRole(msg.sender)||(access.getOwner() == msg.sender), "!A");
        _;
    }
    modifier atLeastProject() {
        require(hasRole(PROJECT_ROLE,msg.sender)||access.hasAdminRole(msg.sender)||(access.getOwner() == msg.sender), "!A");
        _;
    }
    /// @notice withdrawERC20Token
    /// @param token, address of ERC20
    /// @param amount, withdraw amount of ERC20
    function withdrawERC20Token(address token,uint256 amount) public onlyOwner {
        IERC20(token).safeTransfer(msg.sender, amount);
    }
    function addToProjectRole(address account) public atLeastAdmin {
        _grantRole(PROJECT_ROLE, account);
        emit AddProjectRoleEvent(msg.sender,account);
    }
    function removeFromProjectRole(address account) public atLeastAdmin {
        _revokeRole(PROJECT_ROLE, account);
        emit removeProjectRoleEvent(msg.sender,account);
    }
    /// @notice pause all functions
    function pause() public whenNotPaused atLeastAdmin {
        paused = true;
        emit PausedEvent(msg.sender);
    }
    /// @notice unpause all functions
    function unpause() public whenPaused atLeastAdmin{
        paused = false;
        emit UnPausedEvent(msg.sender);
    }
    /// @notice stopStake,stop stake immediately or stop at plan block number
    /// @param _stop, true for stop immediately
    /// @param _stopAtBlock, _stop for stop at block _stopAtBlock
    function stopStake(bool _stop,uint256 _stopAtBlock) public atLeastAdmin{
        stopped = _stop;
        stopAtBlock=_stopAtBlock;
        emit StopEvent(msg.sender);
    }
    /// @notice setEachBlockSecond
    /// @param second, second of one block for this blockchain
    function setEachBlockSecond(uint256 second) public atLeastAdmin{
        uint256 origin=eachBlockSecond;
        eachBlockSecond=second;
        emit SetEachBlockSecondEvent(msg.sender,origin,eachBlockSecond);
    }
    /// @notice setAllowInstantUnbound
    /// @param allow, true for allow user to instant unbound
    function setAllowInstantUnbound(bool allow) public atLeastAdmin{
        allowedInstantUnbound=allow;
        emit SetAllowInstantUnboundEvent(msg.sender,allow);
    }
    /// @notice setYearBlock
    /// @param number, block count of this blockchain for 1 year
    function setYearBlock(uint256 number) public atLeastAdmin{
        uint256 origin=yearBlock;
        yearBlock=number;
        emit SetYearBlockEvent(msg.sender,origin,yearBlock);
    }
    /// @notice setFeeAddress
    /// @param account, address to receive fees
    function setFeeAddress(address account) public atLeastAdmin{
        address origin=feeAddress;
        feeAddress=account;
        emit SetFeeAddressEvent(msg.sender,origin,feeAddress);
    }
    /// @notice setCommission
    /// @param commission, range is [0,10000] ,1728 means 17.28%
    function setCommission(uint256 commission) public atLeastAdmin{
        require(commission<=10000);
        uint256 origin=commissionPercentage;
        commissionPercentage=commission;
        emit SetCommissionEvent(msg.sender,origin,commissionPercentage);
    }
    /// @notice changeUserWhitelistSwitch
    /// @param switchFlag, true for open user whitelist check
    function changeUserWhitelistSwitch(bool switchFlag) public atLeastProject {
        bool origin=userWhitelistSwitchOpened;
        userWhitelistSwitchOpened=switchFlag;
        emit ChangeUserWhitelistSwitchEvent(msg.sender,origin,userWhitelistSwitchOpened);
    }
    /// @notice addToUserWhitelist
    /// @param user, stake user address
    function addToUserWhitelist(address user) public atLeastProject {
        userWhitelist[user] = true;
        emit AddUserWhitelistEvent(msg.sender,user);
    }
    /// @notice removeFromUserWhitelist
    /// @param user, stake user address
    function removeFromUserWhitelist(address user) public atLeastProject {
        userWhitelist[user] = false;
        emit RemoveUserWhitelistEvent(msg.sender,user);
    }
    /// @notice setMinStakeAmount,please notice this amount is important! please keeping this minAmount with a suitable USD value to avoid array length attacks
    /// @param minAmount, stake token amount
    function setMinStakeAmount(uint256 minAmount) public atLeastProject{
        uint256 origin=minStakeAmount;
        minStakeAmount=minAmount;
        emit SetMinStakeAmountEvent(msg.sender,origin,minStakeAmount);
    }
    /// @notice setUnboundPeriodBlock
    /// @param number, how many block user can redeem back their principal
    function setUnboundPeriodBlock(uint256 number) public atLeastProject{
        uint256 origin=unboundPeriodBlock;
        unboundPeriodBlock=number;
        emit SetUnboundPeriodBlockEvent(msg.sender,origin,unboundPeriodBlock);
    }
    /// @notice setPayoutInterval
    /// @param number, claim reward interval block number
    function setPayoutInterval(uint256 number) public atLeastProject{
        uint256 origin=payoutIntervalBlock;
        payoutIntervalBlock=number;
        emit SetPayoutIntervalBlockEvent(msg.sender,origin,payoutIntervalBlock);
    }
    /// @notice setInstantUnboundFee
    /// @param feePer, instant unbound fee from principal,range is [0,10000],1728 means 17.28%
    function setInstantUnboundFee(uint256 feePer) public atLeastAdmin{
        require(feePer<=10000);
        uint256 origin=instantUnboundFeePercentage;
        instantUnboundFeePercentage=feePer;
        emit SetInstantUnboundFeeEvent(msg.sender,origin,instantUnboundFeePercentage);
    }
    constructor(Structs.StakingParams memory params) {
        stakingToken=params.stakingToken;
        minStakeAmount=params.minStakeAmount;
        payoutIntervalBlock=params.payoutIntervalBlock;
        unboundPeriodBlock=params.unboundPeriodBlock;
        instantUnboundFeePercentage=params.instantUnboundFeePercentage;
        access= IAdminAccess(params.accessControl);
        feeAddress=params.feeAddress;
        commissionPercentage=params.commissionPercentage;
        stopped=false;
        stopAtBlock=block.number+100*yearBlock;
        allowedInstantUnbound=params.allowedInstantUnbound;
    }

}
// File: @openzeppelin/contracts/utils/structs/EnumerableSet.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/structs/EnumerableSet.sol)

pragma solidity ^0.8.0;

/**
 * @dev Library for managing
 * https://en.wikipedia.org/wiki/Set_(abstract_data_type)[sets] of primitive
 * types.
 *
 * Sets have the following properties:
 *
 * - Elements are added, removed, and checked for existence in constant time
 * (O(1)).
 * - Elements are enumerated in O(n). No guarantees are made on the ordering.
 *
 * ```
 * contract Example {
 *     // Add the library methods
 *     using EnumerableSet for EnumerableSet.AddressSet;
 *
 *     // Declare a set state variable
 *     EnumerableSet.AddressSet private mySet;
 * }
 * ```
 *
 * As of v3.3.0, sets of type `bytes32` (`Bytes32Set`), `address` (`AddressSet`)
 * and `uint256` (`UintSet`) are supported.
 *
 * [WARNING]
 * ====
 *  Trying to delete such a structure from storage will likely result in data corruption, rendering the structure unusable.
 *  See https://github.com/ethereum/solidity/pull/11843[ethereum/solidity#11843] for more info.
 *
 *  In order to clean an EnumerableSet, you can either remove all elements one by one or create a fresh instance using an array of EnumerableSet.
 * ====
 */
library EnumerableSet {
    // To implement this library for multiple types with as little code
    // repetition as possible, we write it in terms of a generic Set type with
    // bytes32 values.
    // The Set implementation uses private functions, and user-facing
    // implementations (such as AddressSet) are just wrappers around the
    // underlying Set.
    // This means that we can only create new EnumerableSets for types that fit
    // in bytes32.

    struct Set {
        // Storage of set values
        bytes32[] _values;
        // Position of the value in the `values` array, plus 1 because index 0
        // means a value is not in the set.
        mapping(bytes32 => uint256) _indexes;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function _add(Set storage set, bytes32 value) private returns (bool) {
        if (!_contains(set, value)) {
            set._values.push(value);
            // The value is stored at length-1, but we add 1 to all indexes
            // and use 0 as a sentinel value
            set._indexes[value] = set._values.length;
            return true;
        } else {
            return false;
        }
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function _remove(Set storage set, bytes32 value) private returns (bool) {
        // We read and store the value's index to prevent multiple reads from the same storage slot
        uint256 valueIndex = set._indexes[value];

        if (valueIndex != 0) {
            // Equivalent to contains(set, value)
            // To delete an element from the _values array in O(1), we swap the element to delete with the last one in
            // the array, and then remove the last element (sometimes called as 'swap and pop').
            // This modifies the order of the array, as noted in {at}.

            uint256 toDeleteIndex = valueIndex - 1;
            uint256 lastIndex = set._values.length - 1;

            if (lastIndex != toDeleteIndex) {
                bytes32 lastValue = set._values[lastIndex];

                // Move the last value to the index where the value to delete is
                set._values[toDeleteIndex] = lastValue;
                // Update the index for the moved value
                set._indexes[lastValue] = valueIndex; // Replace lastValue's index to valueIndex
            }

            // Delete the slot where the moved value was stored
            set._values.pop();

            // Delete the index for the deleted slot
            delete set._indexes[value];

            return true;
        } else {
            return false;
        }
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function _contains(Set storage set, bytes32 value) private view returns (bool) {
        return set._indexes[value] != 0;
    }

    /**
     * @dev Returns the number of values on the set. O(1).
     */
    function _length(Set storage set) private view returns (uint256) {
        return set._values.length;
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function _at(Set storage set, uint256 index) private view returns (bytes32) {
        return set._values[index];
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function _values(Set storage set) private view returns (bytes32[] memory) {
        return set._values;
    }

    // Bytes32Set

    struct Bytes32Set {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(Bytes32Set storage set, bytes32 value) internal returns (bool) {
        return _add(set._inner, value);
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(Bytes32Set storage set, bytes32 value) internal returns (bool) {
        return _remove(set._inner, value);
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(Bytes32Set storage set, bytes32 value) internal view returns (bool) {
        return _contains(set._inner, value);
    }

    /**
     * @dev Returns the number of values in the set. O(1).
     */
    function length(Bytes32Set storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function at(Bytes32Set storage set, uint256 index) internal view returns (bytes32) {
        return _at(set._inner, index);
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function values(Bytes32Set storage set) internal view returns (bytes32[] memory) {
        return _values(set._inner);
    }

    // AddressSet

    struct AddressSet {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(AddressSet storage set, address value) internal returns (bool) {
        return _add(set._inner, bytes32(uint256(uint160(value))));
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(AddressSet storage set, address value) internal returns (bool) {
        return _remove(set._inner, bytes32(uint256(uint160(value))));
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(AddressSet storage set, address value) internal view returns (bool) {
        return _contains(set._inner, bytes32(uint256(uint160(value))));
    }

    /**
     * @dev Returns the number of values in the set. O(1).
     */
    function length(AddressSet storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function at(AddressSet storage set, uint256 index) internal view returns (address) {
        return address(uint160(uint256(_at(set._inner, index))));
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function values(AddressSet storage set) internal view returns (address[] memory) {
        bytes32[] memory store = _values(set._inner);
        address[] memory result;

        /// @solidity memory-safe-assembly
        assembly {
            result := store
        }

        return result;
    }

    // UintSet

    struct UintSet {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(UintSet storage set, uint256 value) internal returns (bool) {
        return _add(set._inner, bytes32(value));
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(UintSet storage set, uint256 value) internal returns (bool) {
        return _remove(set._inner, bytes32(value));
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(UintSet storage set, uint256 value) internal view returns (bool) {
        return _contains(set._inner, bytes32(value));
    }

    /**
     * @dev Returns the number of values on the set. O(1).
     */
    function length(UintSet storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function at(UintSet storage set, uint256 index) internal view returns (uint256) {
        return uint256(_at(set._inner, index));
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function values(UintSet storage set) internal view returns (uint256[] memory) {
        bytes32[] memory store = _values(set._inner);
        uint256[] memory result;

        /// @solidity memory-safe-assembly
        assembly {
            result := store
        }

        return result;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly
                /// @solidity memory-safe-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/draft-IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/draft-IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/draft-IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/security/Pausable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (security/Pausable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        _requireNotPaused();
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        _requirePaused();
        _;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Throws if the contract is paused.
     */
    function _requireNotPaused() internal view virtual {
        require(!paused(), "Pausable: paused");
    }

    /**
     * @dev Throws if the contract is not paused.
     */
    function _requirePaused() internal view virtual {
        require(paused(), "Pausable: not paused");
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: @openzeppelin/contracts/access/IAccessControlEnumerable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/IAccessControlEnumerable.sol)

pragma solidity ^0.8.0;

import "./IAccessControl.sol";

/**
 * @dev External interface of AccessControlEnumerable declared to support ERC165 detection.
 */
interface IAccessControlEnumerable is IAccessControl {
    /**
     * @dev Returns one of the accounts that have `role`. `index` must be a
     * value between 0 and {getRoleMemberCount}, non-inclusive.
     *
     * Role bearers are not sorted in any particular way, and their ordering may
     * change at any point.
     *
     * WARNING: When using {getRoleMember} and {getRoleMemberCount}, make sure
     * you perform all queries on the same block. See the following
     * https://forum.openzeppelin.com/t/iterating-over-elements-on-enumerableset-in-openzeppelin-contracts/2296[forum post]
     * for more information.
     */
    function getRoleMember(bytes32 role, uint256 index) external view returns (address);

    /**
     * @dev Returns the number of accounts that have `role`. Can be used
     * together with {getRoleMember} to enumerate all bearers of a role.
     */
    function getRoleMemberCount(bytes32 role) external view returns (uint256);
}

// File: @openzeppelin/contracts/access/IAccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/IAccessControl.sol)

pragma solidity ^0.8.0;

/**
 * @dev External interface of AccessControl declared to support ERC165 detection.
 */
interface IAccessControl {
    /**
     * @dev Emitted when `newAdminRole` is set as ``role``'s admin role, replacing `previousAdminRole`
     *
     * `DEFAULT_ADMIN_ROLE` is the starting admin for all roles, despite
     * {RoleAdminChanged} not being emitted signaling this.
     *
     * _Available since v3.1._
     */
    event RoleAdminChanged(bytes32 indexed role, bytes32 indexed previousAdminRole, bytes32 indexed newAdminRole);

    /**
     * @dev Emitted when `account` is granted `role`.
     *
     * `sender` is the account that originated the contract call, an admin role
     * bearer except when using {AccessControl-_setupRole}.
     */
    event RoleGranted(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Emitted when `account` is revoked `role`.
     *
     * `sender` is the account that originated the contract call:
     *   - if using `revokeRole`, it is the admin role bearer
     *   - if using `renounceRole`, it is the role bearer (i.e. `account`)
     */
    event RoleRevoked(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) external view returns (bool);

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {AccessControl-_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) external view returns (bytes32);

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been granted `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) external;
}

// File: @openzeppelin/contracts/access/AccessControlEnumerable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (access/AccessControlEnumerable.sol)

pragma solidity ^0.8.0;

import "./IAccessControlEnumerable.sol";
import "./AccessControl.sol";
import "../utils/structs/EnumerableSet.sol";

/**
 * @dev Extension of {AccessControl} that allows enumerating the members of each role.
 */
abstract contract AccessControlEnumerable is IAccessControlEnumerable, AccessControl {
    using EnumerableSet for EnumerableSet.AddressSet;

    mapping(bytes32 => EnumerableSet.AddressSet) private _roleMembers;

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IAccessControlEnumerable).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev Returns one of the accounts that have `role`. `index` must be a
     * value between 0 and {getRoleMemberCount}, non-inclusive.
     *
     * Role bearers are not sorted in any particular way, and their ordering may
     * change at any point.
     *
     * WARNING: When using {getRoleMember} and {getRoleMemberCount}, make sure
     * you perform all queries on the same block. See the following
     * https://forum.openzeppelin.com/t/iterating-over-elements-on-enumerableset-in-openzeppelin-contracts/2296[forum post]
     * for more information.
     */
    function getRoleMember(bytes32 role, uint256 index) public view virtual override returns (address) {
        return _roleMembers[role].at(index);
    }

    /**
     * @dev Returns the number of accounts that have `role`. Can be used
     * together with {getRoleMember} to enumerate all bearers of a role.
     */
    function getRoleMemberCount(bytes32 role) public view virtual override returns (uint256) {
        return _roleMembers[role].length();
    }

    /**
     * @dev Overload {_grantRole} to track enumerable memberships
     */
    function _grantRole(bytes32 role, address account) internal virtual override {
        super._grantRole(role, account);
        _roleMembers[role].add(account);
    }

    /**
     * @dev Overload {_revokeRole} to track enumerable memberships
     */
    function _revokeRole(bytes32 role, address account) internal virtual override {
        super._revokeRole(role, account);
        _roleMembers[role].remove(account);
    }
}

// File: @openzeppelin/contracts/access/AccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/AccessControl.sol)

pragma solidity ^0.8.0;

import "./IAccessControl.sol";
import "../utils/Context.sol";
import "../utils/Strings.sol";
import "../utils/introspection/ERC165.sol";

/**
 * @dev Contract module that allows children to implement role-based access
 * control mechanisms. This is a lightweight version that doesn't allow enumerating role
 * members except through off-chain means by accessing the contract event logs. Some
 * applications may benefit from on-chain enumerability, for those cases see
 * {AccessControlEnumerable}.
 *
 * Roles are referred to by their `bytes32` identifier. These should be exposed
 * in the external API and be unique. The best way to achieve this is by
 * using `public constant` hash digests:
 *
 * ```
 * bytes32 public constant MY_ROLE = keccak256("MY_ROLE");
 * ```
 *
 * Roles can be used to represent a set of permissions. To restrict access to a
 * function call, use {hasRole}:
 *
 * ```
 * function foo() public {
 *     require(hasRole(MY_ROLE, msg.sender));
 *     ...
 * }
 * ```
 *
 * Roles can be granted and revoked dynamically via the {grantRole} and
 * {revokeRole} functions. Each role has an associated admin role, and only
 * accounts that have a role's admin role can call {grantRole} and {revokeRole}.
 *
 * By default, the admin role for all roles is `DEFAULT_ADMIN_ROLE`, which means
 * that only accounts with this role will be able to grant or revoke other
 * roles. More complex role relationships can be created by using
 * {_setRoleAdmin}.
 *
 * WARNING: The `DEFAULT_ADMIN_ROLE` is also its own admin: it has permission to
 * grant and revoke this role. Extra precautions should be taken to secure
 * accounts that have been granted it.
 */
abstract contract AccessControl is Context, IAccessControl, ERC165 {
    struct RoleData {
        mapping(address => bool) members;
        bytes32 adminRole;
    }

    mapping(bytes32 => RoleData) private _roles;

    bytes32 public constant DEFAULT_ADMIN_ROLE = 0x00;

    /**
     * @dev Modifier that checks that an account has a specific role. Reverts
     * with a standardized message including the required role.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     *
     * _Available since v4.1._
     */
    modifier onlyRole(bytes32 role) {
        _checkRole(role);
        _;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IAccessControl).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) public view virtual override returns (bool) {
        return _roles[role].members[account];
    }

    /**
     * @dev Revert with a standard message if `_msgSender()` is missing `role`.
     * Overriding this function changes the behavior of the {onlyRole} modifier.
     *
     * Format of the revert message is described in {_checkRole}.
     *
     * _Available since v4.6._
     */
    function _checkRole(bytes32 role) internal view virtual {
        _checkRole(role, _msgSender());
    }

    /**
     * @dev Revert with a standard message if `account` is missing `role`.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     */
    function _checkRole(bytes32 role, address account) internal view virtual {
        if (!hasRole(role, account)) {
            revert(
                string(
                    abi.encodePacked(
                        "AccessControl: account ",
                        Strings.toHexString(uint160(account), 20),
                        " is missing role ",
                        Strings.toHexString(uint256(role), 32)
                    )
                )
            );
        }
    }

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) public view virtual override returns (bytes32) {
        return _roles[role].adminRole;
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     *
     * May emit a {RoleGranted} event.
     */
    function grantRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _grantRole(role, account);
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     *
     * May emit a {RoleRevoked} event.
     */
    function revokeRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _revokeRole(role, account);
    }

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been revoked `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     *
     * May emit a {RoleRevoked} event.
     */
    function renounceRole(bytes32 role, address account) public virtual override {
        require(account == _msgSender(), "AccessControl: can only renounce roles for self");

        _revokeRole(role, account);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event. Note that unlike {grantRole}, this function doesn't perform any
     * checks on the calling account.
     *
     * May emit a {RoleGranted} event.
     *
     * [WARNING]
     * ====
     * This function should only be called from the constructor when setting
     * up the initial roles for the system.
     *
     * Using this function in any other way is effectively circumventing the admin
     * system imposed by {AccessControl}.
     * ====
     *
     * NOTE: This function is deprecated in favor of {_grantRole}.
     */
    function _setupRole(bytes32 role, address account) internal virtual {
        _grantRole(role, account);
    }

    /**
     * @dev Sets `adminRole` as ``role``'s admin role.
     *
     * Emits a {RoleAdminChanged} event.
     */
    function _setRoleAdmin(bytes32 role, bytes32 adminRole) internal virtual {
        bytes32 previousAdminRole = getRoleAdmin(role);
        _roles[role].adminRole = adminRole;
        emit RoleAdminChanged(role, previousAdminRole, adminRole);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * Internal function without access restriction.
     *
     * May emit a {RoleGranted} event.
     */
    function _grantRole(bytes32 role, address account) internal virtual {
        if (!hasRole(role, account)) {
            _roles[role].members[account] = true;
            emit RoleGranted(role, account, _msgSender());
        }
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * Internal function without access restriction.
     *
     * May emit a {RoleRevoked} event.
     */
    function _revokeRole(bytes32 role, address account) internal virtual {
        if (hasRole(role, account)) {
            _roles[role].members[account] = false;
            emit RoleRevoked(role, account, _msgSender());
        }
    }
}